SubshiftsEigenvalues

February 26, 2024

[1]:
```
# load the package that allows to compute eigenvalues
from eigenmorphic import *
```

1 Thue-Morse

[2]:
```
s = WordMorphism('a->ab,b->ba')
morphic_eigenvalues(s)
```

[2]:
```
1/2Z[1/2]
```

2 Regular paperfolding

[3]:
```
s = WordMorphism('a->ab,b->cb,c->ad,d->cd')
t = WordMorphism('a->11,b->01,c->10,d->00')
t(s.fixed_point('a'))
```

[3]:
```
word: 11011001110010011101100011001001110110011…
```

[4]:
```
morphic_eigenvalues(s, t)
```

[4]:
```
1/8Z[1/2]
```

3 Constant-length with coboundary

[5]:
```
s = WordMorphism('a->aca,b->acb,c->cbc')
coboundary_basis(s) # basis of the space of coboundaries
```

[5]:
```
[ 1  1 -1]
```

[6]:
```
show(morphic_eigenvalues(s))
```

$$\frac{1}{2}\mathbb{Z}\left[\frac{1}{3}\right]$$

4 Weakly mixing with coboundary

[7]:
```
s = WordMorphism('a->d,b->ca,c->bd,d->abc')
coboundary_basis(s)
```

[7]:
```
[ 0  1 -1  0]
```

[8]:
```
m = s.incidence_matrix()
m.charpoly().factor()
```

[8]:
```
(x - 2) * x * (x + 1)^2
```

[9]:
```
morphic_eigenvalues(s)
```

[9]:
```
Integer Ring
```

5 Unimodular with coboundary and eigenvalues

[10]:
```
s = WordMorphism('a->aba,b->cb,c->cba')
coboundary_basis(s)
```

[10]:
```
[ 1 -1  0]
```

[11]:
```
m = s.incidence_matrix()
m.charpoly().factor()
```

[11]:
```
(x - 1) * (x^2 - 3*x + 1)
```

[12]:
```
morphic_eigenvalues(s)
```

[12]:
```
Z*{1, b}
where b is root of x^2 - 3*x + 1
```

6 Reducible substitution with non-trivial graph of B

[13]:
```
s = WordMorphism('1->16,2->122,3->12,4->3,5->124,6->15')
m = s.incidence_matrix()
print(m.charpoly().factor())
eig = morphic_eigenvalues(s)
eig
```

```
(x^3 - 3*x^2 + 2*x - 1) * (x^3 - x - 1)
```

[13]:
```
Z*{1, 3*b, 3*b^2}
where b is root of x^3 - 3*x^2 + 2*x - 1
```

7 Irreducible substitution with non-trivial graph of B (ex 2 in [FMN])

[14]:
```
s = WordMorphism('a->abdd,b->bc,c->d,d->a')
eig = morphic_eigenvalues(s)
eig
```

[14]:
```
Z*{1, -b^2 + b}
where b is root of x^4 - 2*x^3 - x^2 + 2*x - 1
```

[15]:
```
eig.b.parent().galois_group()
```

[15]:
```
Galois group 4T3 (D(4)) with order 8 of x^4 - 2*x^3 - x^2 + 2*x - 1
```

[16]:
```
m = s.incidence_matrix()
c,b,_,_ = m.eigenvalues()
c == 1-b
```

[16]:
```
True
```

8 Example with coboundary space of dimension 2

[17]:
```
s = WordMorphism('a->b,b->cdad,c->abcb,d->c')
m = s.incidence_matrix()
print(m.charpoly().factor())
```

```
(x^2 - 2*x - 2) * (x^2 + x + 1)
```

[18]:
```
coboundary_basis(s)
```

[18]:
```
[ 1 -1  1 -1]
[ 0  1 -1  0]
```

[19]:
```
morphic_eigenvalues(s)
```

[19]:
```
Z[1/b]*{1/3*b + 1/6, 1/2*b}
where b is root of x^2 - 2*x - 2
```

9 Family of weakly mixing IET, example 4.3.1 in [DS16]

[20]:
```
k = 2 # choose k
str = "10101"+("0"*(k-1))+"100111000011111100000"+("1"*(k-1))+"0"
s = rauzy_loop_substitution("7654321", str)
s
```

[20]:
```
WordMorphism: 1->1617, 2->16252617, 3->162534352617, 4->1625344352617,
5->1625352617, 6->1626162617, 7->162617
```

[21]:
```
m = s.incidence_matrix()
m.charpoly().factor()
```

[21]:
```
(x - 1) * (x^3 - 8*x^2 + 6*x - 1) * (x^3 - 6*x^2 + 8*x - 1)
```

[22]:
```
morphic_eigenvalues(s)
```

[22]:
```
Integer Ring
```

10 Arnoux-Yoccoz

[23]:
```
pi = x^3 + x^2 + x - 1
b = max(pi.roots(ring=AA))[0]
v = [-1/2*b + 1/2, b - 1/2, 1/2*b, 1/2*b^2, 1/2*b^2, -1/2*b^2 - 1/2*b + 1/2, -1/
 ↪2*b^2 - 1/2*b + 1/2]
# find substitutions describing the shift
t, s = rauzy_loop_substitution("2547631", v, get_preperiod=1)
t, s
```

[23]:
```
(WordMorphism: 1->15, 2->2, 3->1734335, 4->163434, 5->1635, 6->1634335, 7->1734,
 WordMorphism: 1->15172, 2->172, 3->1734365172, 4->15643472, 5->1565172,
6->1564365172, 7->173472)
```

[24]:
```
m = s.incidence_matrix()
m.charpoly().factor()
```

[24]:
```
(x - 1) * (x^3 - 7*x^2 + 5*x - 1) * (x^3 - 5*x^2 + 7*x - 1)
```

[25]:
```
# eigenvalues of the IET
morphic_eigenvalues(s, t)
```

[25]:
```
Z*{1, b, b^2}
where b is root of x^3 - 7*x^2 + 5*x - 1
```

[26]:
```
# eigenvalues of the subshift of s (without the pre-period)
morphic_eigenvalues(s)
```

[26]:
```
Integer Ring
```

11 Example coming from Novikov problem

[27]:
```
# list S of substitutions whose S-adic subshift is conjugate to a minimal CET4
ls = [WordMorphism('0->0,1->1,2->361,3->3,4->4,5->5,6->527,7->7'),
     WordMorphism('0->527,1->520,2->0,3->1,4->361,5->461,6->4,7->5'),
     WordMorphism('0->361,1->3,2->4,3->427,4->527,5->7,6->0,7->360'),
     WordMorphism('0->4,1->427,2->527,3->520,4->0,5->360,6->361,7->461'),
     WordMorphism('0->0,1->052,2->2,3->3,4->4,5->614,6->6,7->7'),
     WordMorphism('0->0,1->1,2->2,3->274,4->4,5->5,6->6,7->036')]
```

[28]:
```
# choose a random product of ls
s = product([choice(ls) for _ in range(5)])
# choose another random product of ls
t = product([choice(ls) for _ in range(5)])
#s = ls[1]
print(s.is_primitive())
t, s
```

```
True
```

[28]:
```
(WordMorphism: 0->427036174361427036174360527035274360527036…,
1->42703617436142703617436052703527436052703527436…,
4->74361427036174361427035274360527035274360…,
5->3614270361743614270352743605270352743605270…, 6->361, 7->4270361,
 WordMorphism: 0->460520614, 1->4605206143605206146142705205246142705205246142…,
2->614360520614614270520524614270520524614…, 3->61436052061461427052052461420,
4->052461420, 5->360520614614360520524614270520524614…,
6->3605206146143605205246142705205246142…, 7->4605206146143605205246142705205246142…)
```

[29]:
```
# eigenvalues of the subshift with a pre-period
morphic_eigenvalues(s, t, check_periodic=False, check_recognizable=False)
```

[29]:
```
Z*{43453/653730*b^2 + 6319/65373*b + 1/653730, 5/6*b^2 + 1/6*b, b^2}
where b is root of x^3 - 23*x^2 - 13*x - 1
```

[30]:
```
# eigenvalues of the subshift without a pre-period
morphic_eigenvalues(s, check_periodic=False)
```

[30]:
```
Z*{3/10*b^2 + 1/10, 1/2*b^2 + 1/2*b, b^2}
where b is root of x^3 - 23*x^2 - 13*x - 1
```

12 Example with a big coboundary

[31]:
```
s = WordMorphism('a->h,b->i,c->lbi,d->akc,e->c,f->dkc,g->ekc,h->fbi,i->fbj,j->f,k->g,l->gkc')
m = s.incidence_matrix()
```

```
print(m.charpoly().factor())
coboundary_basis(s)
```

```
(x - 1) * (x + 1)^4 * (x^2 - x + 1) * (x^2 + 1) * (x^3 - 2*x^2 - 1)
```

[31]:
```
[ 0  0  0  0 -1 -1  0  0  1  0  0  0]
[ 0  1  1  0  0  0 -1  0 -1 -1  0  0]
[ 0  0 -1 -1  0  0  0  0  0  1  1  0]
```

[32]:
```
morphic_eigenvalues(s)
```

[32]:
```
Z*{1/2*b^2 + 1/2, 1/2*b^2 + 1/2*b, b^2}
where b is root of x^3 - 2*x^2 - 1
```

13 Example 1 in [FMN]

[33]:
```
s = WordMorphism('a->abbbccccccccccddddddddd,b->bccc,c->d,d->a')
morphic_eigenvalues(s)
```

[33]:
```
Z*{-1/3*b^3 + 4/3*b^2 + 2/3*b + 1/3, -b^3 + 4*b^2 + 2*b}
where b is root of x^4 - 2*x^3 - 7*x^2 - 2*x + 1
```